SSIE 553: Project Report

Christopher Woloshyn

October 28, 2019

1 Introduction

This report contains a comprehensive analysis of methodology wherein we seek to find the optimal strategy to minimize the cost of production. In order to solve this problem, we are using a software known as "A Mathematical Programming Language," or AMPL, to construct a model of each of the two production strategies provided. Further, we will be using the cplex solver to find the optimal solutions of our model. After creating each model finding the solutions to each, we will disclose which production strategy is more cost effective, as well as which actions to take in order to optimize this strategy. For the duration of this report, the fixed workforce strategy will be referred to as Strategy 1, and the variable workforce strategy will be referred to as Strategy 2.

2 Problem Formulation

The first step to formulating the problem is to identify the decision variables that will affect the objective function. Since the goal is to minimize cost, every decision variable will be something that influences the total production cost. In both strategy, a time period of variable costs over the course of 6 months will be considered. After this, the next step is to determine the constraints.

For the problem formulation, assume we are considering a model of length M total months, where $M \in \mathbb{Z}$. Now let $m \in \mathbb{Z}$ such that $1 \leq m \leq M$.

2.1 Problem Formulation for Strategy 1

In Strategy 1, there are 4 costs: production cost (per truck), inventory cost (per truck), regular-time labor (per man-month), and overtime labor (per man-month). Let's call these cost parameters c_m^p, c_m^i, c_m^l, and c_m^v, respectively. We want to create variables such that each of these costs correspond with one. This will make the construction of the objective function much cleaner and more organized. We'll call these variables t_m, i_m, l_m, and v_m respectively. Since the workers can produce 1 truck per man-month, the cost of 1 man-month of labor is also the cost (of labor) to produce 1 truck. Hence, it is beneficial to think of costs in terms of trucks rather than labor, for the sake of problem formulation. So, t_m represents the number of trucks produced in month m, which is simply the sum of regular labor and overtime labor, which are represented by l_m and v_m respectively. In other words, $t_m = l_m + v_m$. Lastly, i_m represents the amount of trucks in inventory in month

m. Thus we see that 4 types of costs, each over the course of 6 months yields 24 decision variables.

Defining the objective function is fairly simple since we know what our decision variables are. Each set of M variables corresponds to one cost type: t_m corresponds to production cost, i_m corresponds to inventory cost, l_m corresponds to labor cost, and v_m corresponds to overtime cost. So, our objective function z is the sum over every month, of each cost type times its corresponding variable.

$$z = \sum_{m=1}^{M} c_m^p t_m + c_m^i i_m + c_m^l l_m + c_m^v v_m \tag{1}$$

We see that in the case $M = 6$, which is our example time period in the problem statement, that there are 24 total decision variables influencing the objective function.

Lastly, we need to define the constraints. There are two types of constraints present in each strategy. The first type I call "simple constraints." These are constraints where a decision variable of a certain month is constrained only by parameters of the same corresponding month. Conversely, "complex constraints" are constraints where a decision variable also depends on a variable from a different month, most commonly the previous month. For Strategy 1, the simple constraints are given in the problem because labor is fixed. So, we take two parameters: L_m and V_m to be the maximum available labor and maximum available overtime labor, respectively. Hence, our simple constraints are:

$$l_m \leq L_m, \tag{2}$$

$$v_m \leq \frac{1}{4} L_m \tag{3}$$

Recall that in the problem statement $\frac{1}{4} L_m$ is given as the maximum available overtime labor.

Second, we need to construct the complex constraints with the inventory. Since inventory affects the total number of trucks that need to be produced per month, it makes sense to consider inventory with demand. We see that without inventory, our constraint with demand is $t_m \leq d_m$, where d_m is the parameter denoting the total demand in month m. The trucks produced in month m, plus the inventory from the previous month $m-1$ should be equal to the demand in month m, plus the inventory for month m (or the excess after meeting the demand). Because we are considering i_{m-1}, we need an initial inventory i_0, which will be some positive integer x. Hence, we set our complex constraint as

$$i_m = i_{m-1} + t_m + d_m, \tag{4}$$

where $i_0 = x$, and $x \in \mathbb{Z}^+$. In the given example, we will use $x = 3$.

2.2 Problem Formulation for Strategy 2

Similarly to Strategy 1, in Strategy 2 we have the same 4 cost types, but also 3 additional cost types: hiring cost (per worker), firing cost (per worker), and back-order cost (per truck) which we will call c^h_m, c^f_m, and c^b_m respectively. Once again we want each of these cost parameters to have a variable associated with it. We will call these corresponding variables h_m, f_m, and b_m. Moreover, since this strategy is of a variable workforce, we also want a variable to keep track of the total number of workers in each month as a function of hiring and firing. We will call this variable w_m, and it will replace the instance of labor cost, because all workers get paid, and the total number of workers will also determine the maximum available labor in a given month. So, by implementing Strategy 2 over $M = 6$ months we have a total of 42 decision variables.

Similarly to Strategy 1, each set of M variables corresponds to one cost type. In addition to the original four, h_m corresponds to hiring cost, f_m corresponds to firing cost, and b_m corresponds to back-order cost. Additionally, w_m corresponds with labor cost, and l_m corresponds to nothing and is not part of the objective function. Again, our objective function z is the sum over every month, of each cost type times its corresponding variable.

$$z = \sum_{m=1}^{M} c^p_m t_m + c^i_m i_m + c^l_m w_m + c^v_m v_m + c^h_m h_m + c^f_m f_m + c^b_m b_m \tag{5}$$

For implementing the constraints into the problem, we will first consider the complex constraint on w_m. w_m is the total number of workers in month m as a function of the number of workers from month $m-1$, plus the number of workers hired in month m, minus the number of workers fired in month m. i.e.

$$w_m = w_{m-1} + h_m - f_m \tag{6}$$

Since we are considering w_{m-1}, we need an initial number of workers w_0, which will be some positive integer y. In the given example, we assume $w_0 = 0$. This is all important for establishing out simple constraints, since l_m depends on w_m

instead of some fixed parameter value like in Strategy 1. Furthermore, v_m also depends on w_m because the maximum amount of overtime is $\frac{1}{4}$ the maximum amount of regular labor. Hence, our simple constraints are

$$l_m \leq w_m, \tag{7}$$

$$v_m \leq \frac{1}{4} w_m \tag{8}$$

Lastly, we need to define the complex constraint on both inventory and back-order, which are both responsible for influencing the demand of a given month. To do this, we will simply add on to our previous inventory constraint. Intuitively, the back-order of month m, b_m, is like the artificial production of trucks. However, since back-order must be fulfilled the following month, b_m is also added to the total demand of month $m+1$. In other words,

$$t_m + i_{m-1} + b_m = d_m + i_m + b_{m-1}, \tag{9}$$

or, putting this equation in terms of inventory,

$$i_m = i_{m-1} + t_m - d_m + b_m - b_{m-1} \tag{10}$$

This is the way this equation is expressed in the code.

And with that, all of the decision variables, objective functions, and constraints have been formulated properly, and are ready for code implementation. In the next section will be a listing of the code from the model, data, and run files for both strategies.

3 AMPL Source Code

Below will be the source code for the .mod files of each strategy. The .dat and .run files are not included here because they have less to do with problem formulation, and more to do with implementing a specific example into the model. Conversly, the .mod files, in conjunction with the problem formulation from Section 2, should provide comprehensive insight into how the problem was implemented into AMPL for the purposes of finding and comparing the optimality of each strategy.

Listing 1: prj_opt_1.mod

```
option solver cplex;
```

```
#create sets
set MONTH;
set MONTH_PLUS_1;

# Parameters for the objective function
param production_cost {m in MONTH};
param inventory_cost {m in MONTH};
param labor_cost {m in MONTH};
param overtime_cost {m in MONTH};

# Parameters for the initial inventory
param initial_inv;

# Parameters for the constraints
param demand {m in MONTH};
param max_labor {m in MONTH};
param max_overtime {m in MONTH};

# Decision variables
var labor {m in MONTH} integer >= 0;
var overtime {m in MONTH} integer >= 0;
var trucks {m in MONTH} = labor[m] + overtime[m];
var inventory {m in MONTH_PLUS_1} integer >=0;

# Objective Function
minimize total_cost: sum {m in MONTH}(
production_cost[m] * trucks[m] +
inventory_cost[m] * inventory[m] +
labor_cost[m] * labor[m] +
overtime_cost[m] * overtime[m]
);

# Constraints on the maximum labor per month
s.t. max_regular_labor {m in MONTH}:
labor[m] <= max_labor[m];

s.t. max_overtime_labor {m in MONTH}:
```

```
overtime[m] <= max_overtime[m];

# Constraints made to update the inventory each month
s.t. set_initial_inv: inventory[0] = initial_inv;
s.t. update_inv {m in MONTH}:
inventory[m] =  inventory[m-1] +
trucks[m] - demand[m];
```

Listing 2: prj_opt_2.mod

```
option solver cplex;

#create sets
set MONTH;
set MONTH_PLUS_1;

# Parameters for the objective function
param production_cost {m in MONTH};
param inventory_cost {m in MONTH};
param labor_cost {m in MONTH};
param overtime_cost {m in MONTH};
param hiring_cost {m in MONTH};
param firing_cost {m in MONTH};
param backorder_cost {m in MONTH};

# Parameters for the initial inventory
param initial_inv;
param initial_back;
param initial_work;

# Parameters for the constraints
param demand {m in MONTH};

# Decision variables
var labor {m in MONTH_PLUS_1} integer >= 0;
var overtime {m in MONTH} integer >= 0;
var trucks {m in MONTH} = labor[m] + overtime[m];
```

```
var inventory {m in MONTH_PLUS_1} integer >=0;

var workers {m in MONTH_PLUS_1} integer >= 0;
var hired {m in MONTH} integer >= 0;
var fired {m in MONTH} integer >= 0;
var backorder {m in MONTH_PLUS_1} integer >= 0;

# Objective Function
minimize total_cost: sum {m in MONTH}(
production_cost[m] * trucks[m] +
inventory_cost[m] * inventory[m] +
labor_cost[m] * workers[m] +
overtime_cost[m] * overtime[m] +
hiring_cost[m] * hired[m] +
firing_cost[m] * fired[m] +
backorder_cost[m] * backorder[m]
);

# Constraints on the maximum labor per month
s.t. max_labor {m in MONTH}: labor[m] <= workers[m];
s.t. max_overtime_labor {m in MONTH}:
overtime[m] <= .25 * workers[m];

# Constraints made to update the inventory each month
s.t. set_initial_inv: inventory[0] = initial_inv;
s.t. set_initial_back: backorder[0] = initial_back;
s.t. update_inv_back {m in MONTH}:
inventory[m] = inventory[m-1] + trucks[m] -
demand[m] + backorder[m] - backorder[m-1];

# Constraints made to update workers each month
s.t. set_initial_work: workers[0] = initial_work;
s.t. update_work {m in MONTH}:
workers[m] = workers[m-1] + hired[m] - fired[m];
```

*Note: some of the comments and spacing were slightly altered for formatting purposes.

4 Results and Conclusion

After implementing the given parameters from the initial problem statement into the .dat files for each strategy, the optimal solutions for both strategies were outputted to the AMPL console. For Strategy 1,

$$z = 20233$$

and for Strategy 2,

$$z = 22839$$

The optimal solution for Strategy 1 is over 2000 units less than the optimal solution for Strategy 2. Hence, it is recommended to implement Strategy 1, the fixed labor model, in order to minimize costs between the two potential models. The best way to minimize costs in Strategy 1 is to follow the guidelines in table 1 below.

Table 1: Solution table for each of the decision variables.

Month	**Labor**	**Overtime**	**Trucks**	**Inventory**	**Demand**
December	-	-	-	3	-
January	120	0	120	23	100
February	130	15	145	68	100
March	120	0	120	38	150
April	150	37	187	25	200
May	100	25	125	0	150
June	100	0	100	0	100

In conclusion, the optimal strategy to minimize costs is to adopt a fixed workforce model instead of a variable workforce model. Moreover, in order to optimize the fixed workforce model, follow the guidelines in table 1 for how much regular and overtime labor to used each month, as well as how many trucks to keep in inventory each month.